

Fiche de travail n°3

Les dimensions inter- et intra-individuelle

Je vous propose dans cette troisième fiche de travail d'estimer, en utilisant tantôt la dimension inter-individuelle tantôt la dimension intra-individuelle, une fonction de production «en taux de croissance» en utilisant les mêmes données que dans la la fiche de travail n°2.

1 La dimension inter-individuelle

Il nous faut dans un premier temps calculer les moyennes individuelles (c'est-à-dire les moyennes par branche) des variables d'intérêt. Le plus simple est d'utiliser la PROC means de SAS. Le fragment de programme suivant s'acquiesce de cela :

```
DATA table ;
  SET table ;
  IF (dlq NE .) AND (dlk NE .) AND (dll NE .) ;
RUN ;
PROC means DATA = table NOPRINT ;
  VAR dlq dlk dll ;
  BY branche ;
  OUTPUT OUT = moyennes MEAN = dlqb dlkb dllb ;
RUN ;
```

Notez qu'il est tout d'abord nécessaire de ne garder que les observations pour lesquelles les variables sont définies. L'option NOPRINT évite d'avoir dans la fenêtre OUTPUT le résultat de la PROC means. L'instruction OUTPUT permet de sauvegarder dans une table SAS des statistiques issues de la PROC means. Le mot-clé MEAN spécifie que l'on cherche à sauvegarder la moyenne par groupe des variables qui ont été indiquées dans l'instruction VAR : il faut à la suite donner le nom que l'on veut attribuer à ces variables.

Le modèle dans la dimension inter-individuelle s'écrit comme suit :

$$(6) \quad \overline{dq}_i = \alpha \overline{dk}_i + \beta \overline{dl}_i + \delta + u_i \quad i = 1, \dots, N$$

en notant \overline{dq}_i la moyenne, pour la branche i , de la variable Δq de terme $q_{it} - q_{it-1}$; de même pour \overline{dk}_i et pour \overline{dl}_i . Pour estimer ce modèle, il suffit de coder :

```
PROC reg DATA = moyennes ;
  MODEL dlqb = dlkb dllb ;
RUN ;
```

2 La dimension intra-individuelle

La dimension intra-individuelle relève de l'écart à la moyenne individuelle. Il faut donc calculer, par exemple, la grandeur $dq_{it} - \overline{dq}_i$. La solution la plus simple repose sur une instruction MERGE pour pouvoir réaliser ce calcul :

```
DATA final ;
  MERGE table moyennes ;
  BY branche ;
  dlqi = dlq - dlqb ;
  dlki = dlk - dlkb ;
  dlli = dll - dllb ;
RUN ;
```

Dans ce fragment de programme, la variable SAS dlq provient de la table table alors que la variable dlqb provient de la table moyennes. Notez l'instruction BY, nécessaire. La table table contient NT observations alors que la table moyennes n'en contient que N.

Rappelons que le modèle qui repose sur l'hypothèse d'effets fixes individuels :

$$(5) \quad dq_{it} = \alpha dk_{it} + \beta dl_{it} + \gamma_i + u_{it} \quad i = 1, \dots, N \quad t = 1, \dots, T$$

est équivalent, en vertu du théorème de FRISCH-WAUGH, au modèle spécifié dans la dimension intra-individuelle :

$$(7) \quad [dq_{it} - \overline{dq}_i] = \alpha [dk_{it} - \overline{dk}_i] + \beta [dl_{it} - \overline{dl}_i] + u_{it} \quad i = 1, \dots, N \quad t = 1, \dots, T$$

Pour estimer ce dernier modèle, il faut coder :

```
PROC reg DATA = final ;
  MODEL dlqi = dlki dlli / NOINT ;
RUN ;
```

Notons que l'on n'obtient pas, pour les écart-types estimés des estimateurs (et en conséquence pour les t de STUDENT), les mêmes valeurs, si l'on utilise la PROC reg sur la table final

```
PROC reg DATA = final ;
  MODEL dlqi = dlki dlli / NOINT ;
RUN ;
```

ou la PROC glm sur la table table

```
PROC glm DATA = table ;
  CLASS branche ;
  MODEL dlq = dlk dll branche / SOLUTION ;
RUN ;
```

Il faut, dans la première régression, introduire une correction de degrés de liberté.

Enfin, la décomposition de la variance totale en la variance inter-individuelle et la variance intra-individuelle s'obtient en utilisant, par exemple, la PROC univariate :

```
PROC univariate DATA = final ;
  VAR dlq dlqb dlqi dlk dlkb dlki dll dllb dlli ;
RUN ;
```